# Maintainer: Justin Davis (juster) <jrcd83@gmail.com>
# $Id: PKGBUILD 71649 2012-05-31 15:44:24Z juster $

pkgname=perl-sub-install
pkgver=0.926
pkgrel=1
pkgdesc='install subroutines into packages easily'
arch=(any)
license=(PerlArtistic GPL)
options=(!emptydirs)
depends=(perl)
url=https://metacpan.org/release/Sub-Install
source=("http://search.cpan.org/CPAN/authors/id/R/RJ/RJBS/Sub-Install-$pkgver.tar.gz")
md5sums=(89a7f82dd840bc2401f281b5f24732b9)
sha512sums=(c536e1ace79ce8c6bd4a6e580cc2e181671b54f07cef231a84094c854779aee556de4b2ff0a95f0e6a8aca87f77a6e19d148071a48fe5c6e9c0ca15314f0c9cc)
_dir="$srcdir/Sub-Install-$pkgver"

build()
(
  cd "$_dir"
  export PERL_MM_USE_DEFAULT=1 PERL_AUTOINSTALL=--skipdeps
  unset PERL5LIB PERL_MM_OPT
  /usr/bin/perl Makefile.PL
  make
)

check()
(
  cd "$_dir"
  export PERL_MM_USE_DEFAULT=1
  unset PERL5LIB
  make test
)

package()
(
  cd "$_dir"
  make install INSTALLDIRS=vendor DESTDIR="$pkgdir"
  find "$pkgdir" -name .packlist -o -name perllocal.pod -delete
)

# Local Variables:
# mode: shell-script
# sh-basic-offset: 2
# End:
# vim:set ts=2 sw=2 et:
